



Harnessing Excel's VBA for Simulating Fundamental Stochastic Processes: A Deep Dive into GBM and CIR Implementations

¹Debasree Goswami

¹Associate Professor, Department of Statistics, Hindu College, University of Delhi, Delhi, India-7

Abstract : This research delves into the application of stochastic processes, specifically the Geometric Brownian Motion (GBM) and the Cox-Ingersoll-Ross (CIR) process, in the realm of quantitative finance. Recognizing the ubiquity of Microsoft Excel in the finance sector, we introduce a hands-on approach to simulate these processes using Visual Basic for Applications (VBA). Our paper highlights the implementation details, ensuring the adaptability of the developed code for various financial stochastic processes. By bridging theoretical finance with practical Excel-based solutions, this research aims to provide a valuable tool for financial analysts and researchers.

Keywords: Stochastic Processes, Geometric Brownian Motion (GBM), Cox-Ingersoll-Ross (CIR) Process, Visual Basic for Applications (VBA), Financial Modeling in Excel, Numerical Simulation, Milstein Scheme.

1. INTRODUCTION

Financial markets are inherently uncertain, with asset prices and interest rates exhibiting random fluctuations over time. To capture and understand these dynamics, stochastic processes, which describe systems evolving over time with a probabilistic element, have become fundamental in the realm of quantitative finance. Among these, the Geometric Brownian Motion (GBM) and the Cox-Ingersoll-Ross (CIR) process stand out due to their widespread applications in modeling stock prices and interest rates, respectively.

While there are numerous software platforms tailored for financial modeling, Microsoft Excel remains a staple in the finance industry, offering accessibility and ease of use. Its embedded programming environment, Visual Basic for Applications (VBA), allows for the creation of custom functions and simulations, making it a versatile tool for financial analysts and researchers. This paper delves into the implementation of the GBM and CIR processes in Excel using VBA, providing a hands-on approach to understanding and simulating these fundamental stochastic processes. Furthermore, we discuss the adaptability of the developed code to model other related processes, exemplified by the ease of transitioning from the CIR to the Ornstein-Uhlenbeck (OU) process.

By bridging the gap between theoretical finance and practical implementation, this paper aims to serve as a valuable resource for both academia and industry professionals looking to harness the power of Excel for stochastic financial modeling.

2. METHODOLOGY

2.1. Stochastic Differential Equation and Milstein Scheme

A general one-dimensional SDE can be written as:

$$dX_t = a(X_t, t)dt + b(X_t, t)dW_t \quad (1)$$

where dX_t is the change in the process X_t , $a(X_t, t)$ is the drift term, $b(X_t, t)$ is the diffusion term, and dW_t is a Wiener process or Brownian motion [1].

The Milstein scheme [1], [2] approximates the solution of the SDE by discretizing time into small intervals of length Δt , and the approximation at time t_{n+1} is given by:

$$X_{t_{n+1}} = X_{t_n} + a(X_{t_n}, t_n)\Delta t + b(X_{t_n}, t_n)\Delta W_n + \frac{1}{2}b(X_{t_n}, t_n)b'(X_{t_n}, t_n)((\Delta W_n)^2 - \Delta t) \quad (2)$$

Here $\Delta W_n = W_{t_{n+1}} - W_{t_n}$ is the increment of the Wiener process, which is normally distributed with mean 0 and variance Δt . Further, $b'(X_{t_n}, t_n)$ is the derivative of the diffusion term with respect to X .

2.2. Geometric Brownian Motion (GBM) and the Milstein Scheme

Geometric Brownian Motion (GBM) is a continuous-time stochastic process [3] widely used in mathematical finance to model stock prices, interest rates, exchange rates, and other financial variables. The process assumes that the percentage changes in the stock price are normally distributed, resulting in a log-normal distribution of the stock price over time. Mathematically, it has the form

$$dS_t = \mu S_t dt + \sigma S_t dW_t. \quad (3)$$

where S_t is the value of the stock price at time t , μ is the drift rate, often interpreted as the expected return of the stock, σ is the volatility coefficient, and W_t is a standard Brownian motion or Wiener process.

The Milstein approximation for GBM is:

$$\Delta S_t = \mu S_t \Delta t + \sigma S_t \Delta W_t + \frac{1}{2} \sigma^2 S_t ((\Delta W_t)^2 - \Delta t) \quad (4)$$

Here, ΔS_t is the change in stock price, Δt is the time step, and ΔW_t is the increment of the Wiener process, which is normally distributed with mean 0 and variance Δt .

The additional term $\frac{1}{2} \sigma^2 S_t ((\Delta W_t)^2 - \Delta t)$ in the Milstein scheme is a correction term which provides an improvement over the Euler-Maruyama method in approximating GBM.

2.3. Cox-Ingersoll-Ross (CIR) Process and the Milstein Scheme

The Cox-Ingersoll-Ross (CIR) process [3] is a type of mean-reverting stochastic process primarily used in finance to model interest rates. The CIR process ensures that interest rates remain positive, making it a more realistic model for certain financial applications compared to other interest rate models.

The stochastic differential equation (SDE) representing the CIR process is given by:

$$dX_t = \alpha(\theta - X_t) dt + \sigma \sqrt{X_t} dW_t. \quad (5)$$

where X_t is the process value at time t , α is the speed of mean reversion, θ is the long-term mean (interest rate), σ is the volatility and dW_t is the increment of a Wiener process.

The Milstein scheme is an extension of the Euler-Maruyama scheme that includes a correction term to account for the second moment of the Wiener process increment. For the CIR process, the Milstein scheme can be written as:

$$\Delta X_t = \alpha(\theta - X_t) \Delta t + \sigma \sqrt{X_t} \Delta W_t + \frac{1}{2} \sigma^2 ((\Delta W_t)^2 - \Delta t) \quad (6)$$

where $\Delta X_t = X_t - X_{t-1}$ is the change in the process value, Δt is the time increment and ΔW_t is a Wiener process increment over the interval Δt , which is normally distributed with mean 0 and variance Δt .

The additional term $\frac{1}{2} \sigma^2 ((\Delta W_t)^2 - \Delta t)$ in the Milstein scheme corrects for the approximation error in the Euler-Maruyama method, making the Milstein scheme a more accurate method for approximating the CIR process.

2.4. Implementation in Excel using VBA

Visual Basic for Applications (VBA) is an event-driven programming language developed by Microsoft. It is primarily used for automating tasks in Microsoft Office applications, with Excel being one of its most popular platforms. VBA allows users to create custom functions, automate repetitive tasks, and develop user-defined models and simulations, making it a powerful tool suitable for financial modeling and analysis.

Further, opting for (*Application.WorksheetFunction.RandBetween*(0, 2^{32}) / (2^{32})) over VBA's native *Rnd*() function provided a more granular distribution of random numbers in the interval [0,1], due to its ability to generate a larger set of distinct values, enhancing the resolution of the random number generation process.

Description of the functions:

MilsteinGBMandEXACT: This function approximates the Geometric Brownian Motion using the Milstein scheme and provides the exact solution values. Depending on the user's selection, it can return either the approximate values (upon selecting one column) or the exact values and approximate values (upon selecting two columns). The function requires the initial stock price (So), drift coefficient (Mu), volatility coefficient (Sigma), and a range of time values (tRange) as its arguments. (Appendix II)

MilsteinCIR: Designed to approximate the Cox-Ingersoll-Ross (CIR) process using the Milstein scheme, this function takes initial parameters such as the initial value (Xo), mean reversion rate (Alpha), long-term mean (Mu), and volatility (Sigma), along with a range of time values (tRange). It then produces an array of approximated values over the specified time frame. (Appendix III)

ValidateParameters: Serving as a utility function, "ValidateParameters" ensures that the input parameters for the above functions are valid. It checks for numeric values, the structure of the time range, and the order of time values, raising errors if any inconsistencies are found. (Appendix I)

3. RESULTS AND DISCUSSION

Using the developed VBA functions, we generated sample paths for both the Geometric Brownian Motion and the Cox-Ingersoll-Ross processes. For a given set of parameters, the functions returned arrays of values representing the evolution of the processes over time. The results were plotted directly within Excel, demonstrating the ease of integration and visualization when using VBA in conjunction with Excel's native charting capabilities. (Appendix IV and V)

Upon comparing the results obtained from our VBA functions with those from other platforms, such as R and Python, we observed a high degree of similarity. This attests to the accuracy of the Milstein scheme's implementation in our VBA functions. While minor discrepancies can arise due to differences in random number generation or numerical methods across platforms, the overall trajectories of the generated paths were consistent.

The Milstein scheme, as implemented in our VBA functions, offers a balance between accuracy and computational efficiency. While there are more sophisticated numerical methods available, [2] the Milstein scheme provides a good trade-off, especially for applications within Excel where rapid computation is often desired.

Furthermore, the flexibility of VBA allows for easy modifications and extensions of these functions, catering to more specific needs or advanced financial models. The integration with Excel also means that users can leverage Excel's data analysis and visualization tools, making the entire modeling and analysis process seamless and efficient.

Further, the Cox-Ingersoll-Ross (CIR) process and the Ornstein–Uhlenbeck (OU) process share structural similarities, with the primary difference being in the volatility term. By adjusting the line 54 in the VBA function “*MilsteinCIR*” (Appendix-III) for the CIR process, one can easily adapt it to model the OU process, demonstrating the flexibility and adaptability of the developed code for various stochastic processes in finance.

4. REFERENCES

- [1] T. Mikosch, “Elementary Stochastic Calculus, with Finance in View,” vol. 6, Oct. 1998, doi: 10.1142/3856.
- [2] P. E. Kloeden and E. Platen, “Numerical Solution of Stochastic Differential Equations,” *Numerical Solution of Stochastic Differential Equations*, 1992, doi: 10.1007/978-3-662-12616-5.
- [3] J. Franke, W. K. Härdle, and C. M. Hafner, “Statistics of Financial Markets,” 2019, doi: 10.1007/978-3-030-13751-9.

ACKNOWLEDGMENT

The author would like to extend heartfelt gratitude to Ms. Jasvinder Goswami, Associate Professor in the Department of Statistics at PGDAV(M) College, University of Delhi, and his wife, for her invaluable discussions and insightful comments on the manuscript.

APPENDIX - I

```

1
2 ' ===== Helper Functions =====
3 Function ValidateParameters(inputs As Variant, tRange As Range) As Boolean
4 ' Check if all inputs are numeric
5 Dim i As Integer
6 For i = LBound(inputs) To UBound(inputs)
7     If Not IsNumeric(inputs(i)) Then
8         Err.Raise 9999, "ValidateParameters", "All input parameters must be numeric values."
9     Exit Function
10 End If
11 Next i
12
13 ' Check if tRange is a single column
14 If tRange.Columns.Count > 1 Then
15     Err.Raise 9999, "ValidateParameters", "tRange must be a single column of values."
16 Exit Function
17 End If
18
19 ' Convert the range to an array
20 Dim t() As Variant
21 t = tRange.Value
22
23 ' Check if tRange contains only numeric values
24 For i = LBound(t, 1) To UBound(t, 1)
25     If Not IsNumeric(t(i, 1)) Then
26         Err.Raise 9999, "ValidateParameters", "tRange must contain only numeric values."
27     Exit Function
28 End If
29 Next i
30
31 ' Check if tRange values are in non-decreasing order
32 For i = LBound(t, 1) + 1 To UBound(t, 1)
33     If t(i, 1) <= t(i - 1, 1) Then
34         Err.Raise 9999, "ValidateParameters", "Values in tRange must be in >= order."
35     Exit Function
36 End If
37 Next i
38
39 ValidateParameters = True
40 End Function
41
42

```

43 APPENDIX - II

```

44 ' ===== Main Functions =====
45 ' Function: ===== MilsteinGBMandEXACT =====
46 ' Purpose: Approximate the GBM using the Milstein scheme and provide the exact solution values.
47 ' The function's behavior changes based on the number of selected columns:
48 ' - 2 columns: Returns exact values in the first column and approximate values in the second.
49 ' - 1 column: Returns only the approximate values.
50 ' Parameters:
51 ' - So: Initial stock price
52 ' - Mu: Drift coefficient
53 ' - Sigma: Volatility coefficient
54 ' - tRange: Range of time values
55 ' Returns: An array of exact and/or approximate stock prices based on the number of selected columns.
56
57 Function MilsteinGBMandEXACT(So As Variant, Mu As Variant, Sigma As Variant, tRange As Range) As Variant()
58
59 On Error GoTo ErrorHandler
60
61 Dim DeltaT As Double
62 Dim CumDeltaW As Double
63 Dim DeltaW As Double
64 Dim PrevSt As Double
65 Dim ForNextSt As Double
66 Dim NextSt As Double
67 Dim ExactSt As Double
68 Dim St() As Variant
69 Dim t() As Variant
70 Dim i As Integer
71 Dim n As Integer
72 Dim no_of_column_selected As Integer
73
74 ' Convert the range to an array
75 t = tRange.Value
76
77 ' Determine the number of time steps based on the length of the input time array
78 n = UBound(t, 1) - LBound(t, 1) + 1
79
80 ' Validate the parameters using the helper function
81 If Not ValidateParameters(Array(So, Mu, Sigma), tRange) Then Exit Function
82
83 ' Determine the number of columns selected by the user
84 no_of_column_selected = Application.Caller.Columns.Count
85
86 ' Check the number of columns selected by the user
87 If no_of_column_selected = 2 Then
88 ' User selected two columns
89 ReDim St(1 To n, 1 To 2) As Variant
90 Else
91 ' User selected one column
92 ReDim St(1 To n, 1 To 1) As Variant
93 End If
94
95 ' Initialize the first value
96 St(1, 1) = So
97 If no_of_column_selected = 2 Then
98 St(1, 2) = So
99 End If
100
101 ' Initialize for the first value Wiener Process
102 CumDeltaW = 0
103
104 ' Loop through each time step
105 For i = 2 To n
106 ' Compute the time increment
107 DeltaT = t(i, 1) - t(i - 1, 1)
108
109 ' Generate a random number for the Wiener process increment
110 DeltaW = (Sqr(DeltaT) * Application.NormSInv(Application.WorksheetFunction.RandBetween(0, 2 ^ 32) / (2 ^ 32)))
111
112 CumDeltaW = CumDeltaW + DeltaW
113
114 ' Exact solution for GBM
115 ExactSt = So * Exp((Mu - 0.5 * Sigma ^ 2) * t(i, 1) + Sigma * CumDeltaW)
116
117 ' Milstein scheme for GBM approximation
118 PrevSt = St(i - 1, 1)
119 ForNextSt = Mu * PrevSt * DeltaT + Sigma * PrevSt * DeltaW + 0.5 * (Sigma ^ 2) * PrevSt * ((DeltaW ^ 2) - DeltaT)
120 NextSt = PrevSt + ForNextSt
121
122 ' Assign values to the output array based on the number of selected columns
123 If no_of_column_selected = 2 Then
124 St(i, 1) = ExactSt
125 St(i, 2) = NextSt
126 Else

```

```

127     St(i, 1) = NextSt
128     End If
129     Next i
130
131     ' Return the array of values
132     MilsteinGBMandEXACT = St
133
134     Exit Function
135
136 ErrorHandler:
137     MsgBox "An error occurred in MilsteinGBMandEXACT: " & Err.Description, vbExclamation, "Error"
138 End Function
139

```

1 APPENDIX - III

```

2 ' ===== Main Functions =====
3 ' Function: ===== MilsteinCIR =====
4 ' Purpose: Approximate the Cox-Ingersoll-Ross (CIR) process using the Milstein scheme.
5 ' Parameters:
6 ' - Xo: Initial value
7 ' - Alpha: Speed of mean reversion
8 ' - Mu: Long-term mean
9 ' - Sigma: Volatility coefficient
10 ' - tRange: Range of time values
11 ' Returns: An array of values representing the CIR process.
12
13 Function MilsteinCIR(Xo As Variant, Alpha As Variant, Mu As Variant, Sigma As Variant, tRange As Range) As Variant()
14
15     On Error GoTo ErrorHandler
16
17     Dim n As Integer
18     Dim DeltaW As Double
19     Dim DeltaT As Double
20     Dim PrevSt As Double
21     Dim NextXt0 As Double
22     Dim NextXt1 As Double
23     Dim NextXt2 As Double
24     Dim NextXt3 As Double
25     Dim NextXt As Double
26     Dim Xt() As Variant
27     Dim t() As Variant
28     Dim i As Integer
29
30     ' Convert the range to an array
31     t = tRange.Value
32
33     ' Determine the number of time steps based on the length of the input time array
34     n = UBound(t, 1) - LBound(t, 1) + 1
35
36     ' Validate the parameters using the helper function
37     If Not ValidateParameters(Array(Xo, Alpha, Mu, Sigma), tRange) Then Exit Function
38
39     ' Initialize the St array to store the stock prices
40     ReDim Xt(1 To n, 1 To 1) As Variant
41     Xt(1, 1) = Xo
42
43     ' Loop through each time step to compute the stock price using the Milstein scheme
44     For i = 2 To n
45         ' Compute the time increment
46         DeltaT = t(i, 1) - t(i - 1, 1)
47
48         ' Generate a random number from a standard normal distribution for the Wiener process increment
49         DeltaW = (Sqr(DeltaT) * Application.WorksheetFunction.NormSInv(Application.WorksheetFunction.RandBetween(0, 2 ^ 32) / (2 ^ 32)))
50
51         ' Apply the Milstein scheme formula
52         PrevXt = Xt(i - 1, 1)
53         NextXt = PrevXt + (Alpha * (Mu - PrevXt) * DeltaT) + (Sigma * Sqr(PrevXt) * DeltaW) + (0.25 * (Sigma ^ 2) * DeltaT * ((DeltaW ^ 2) - DeltaT))
54         Xt(i, 1) = NextXt
55     Next i
56
57     ' Return the array of stock prices
58     MilsteinCIR = Xt
59
60     Exit Function
61
62 ErrorHandler:
63     MsgBox "An error occurred in MilsteinCIR: " & Err.Description, vbExclamation, "Error"
64 End Function
65
66

```



APPENDIX - IV

Table - 1			GBM	
		Time (t)	Exact Solution	Approximate Solution
n =	256	0	1.000	1.000
DeltaT: $\Delta t =$	0.00391	0.0039	0.836	0.836
		0.0078	0.924	0.925
mu: $\mu =$	0.01	0.0117	0.904	0.904
Sg: $\sigma =$	2	0.0156	0.734	0.734
$S_0 =$	1	0.0195	0.733	0.733
		0.0234	0.765	0.765
		0.0273	0.690	0.690
		0.0313	0.644	0.643
...
		0.9727	1.273	1.272
		0.9766	1.329	1.329
		0.9805	1.335	1.336
		0.9844	1.378	1.378
		0.9883	1.700	1.700
		0.9922	1.841	1.842
		0.9961	1.684	1.683
		1.0000	1.554	1.554




Fig 1. Equidistant Milstein scheme approximating the Geometric Brownian Motion process for parameter choices in the Table 1.

APPENDIX - V

Table - 2			CIR
		Time (t)	Approximate Solution
n =	128	0	0.5
DeltaT: $\Delta t =$	0.00781	0.0078	0.5191
		0.0156	0.5110
mu: $\mu =$	1	0.0234	0.6123
Sg: $\sigma =$	0.8	0.0313	0.5065
$X_0 =$	0.5	0.0391	0.5785
Alpha: $\alpha =$	5	0.0469	0.4677
		0.0547	0.4184
		0.0625	0.3857
...
		0.9453	1.3250
		0.9531	1.3522
		0.9609	1.3025
		0.9688	1.2873
		0.9766	1.3052
		0.9844	1.3930
		0.9922	1.3613
		1.0000	1.3001




Fig 2. Equidistant Milstein scheme to approximate the Cox-Ingersoll-Ross (CIR) process for parameter choices in the Table 2.